ANCOVA Examples Using SAS

/**************************************

 ANCOVA Using Proc Reg and Proc GLM

 This example illustrates:

      How to import an SPSS file into SAS
      How to create dummy variables for a character variable

      How to create interaction terms

      How to center a variable

      How to create a scatter plot with a regression line for each group

      How to fit an ANCOVA model using Proc Reg

      How to fit an ANCOVA model using Proc GLM

      How to use the Estimate statement in Proc GLM

 Procs used:

      Proc Import
      Proc Contents

      Proc Means

      Proc Freq

      Proc SGplot

      Proc Reg

      Proc GLM

      Proc Univariate

FILENAME: ancova_2009.sas

****************************************/

OPTIONS FORMCHAR="|----|+|---+=|-/\<>*";

options pageno=1 nodate;

title;

We first import the HTWT.SAV dataset from SPSS, get simple descriptive statistics for the numeric variables, and look at oneway frequencies for categorical variables. there are 237 participants who are from 13.9 to 25 years old. It is a cross-sectional study, with each participant having one observation. We can use this data set to examine the relationship of participants’ height to their age and sex.
PROC IMPORT OUT= WORK.htwt 

            DATAFILE= "C:\Documents and Settings\kwelch\Desktop\b510\htwt.sav" 

            DBMS=SAV REPLACE;

RUN;
title "Contents of HTWT Data Set";

proc contents data=htwt;

run;
                         Contents of HTWT Data Set                        

                           The CONTENTS Procedure

Data Set Name        WORK.HTWT                      Observations          237

Member Type          DATA                           Variables             4

Engine               V9                             Indexes               0

Created              Thursday, April 16,            Observation Length    32

                     2009 08:08:45 AM

Last Modified        Thursday, April 16,            Deleted Observations  0

                     2009 08:08:45 AM

Protection                                          Compressed            NO

Data Set Type                                       Sorted                NO

Label

Data Representation  WINDOWS_32

Encoding             wlatin1  Western (Windows)

                      Engine/Host Dependent Information

Data Set Page Size          4096

Number of Data Set Pages    3

First Data Page             1

Max Obs per Page            126

Obs in First Data Page      82

Number of Data Set Repairs  0

Filename                    C:\DOCUME~1\kwelch\LOCALS~1\Temp\SAS

                            Temporary Files\_TD4720\htwt.sas7bdat

Release Created             9.0201M0

Host Created                XP_PRO

                 Alphabetic List of Variables and Attributes

        #    Variable    Type    Len    Format    Informat    Label

        2    AGE         Num       8    F5.2                  AGE

        3    HEIGHT      Num       8    F5.2                  HEIGHT

        1    SEX         Char      8    $8.       $8.         SEX

        4    WEIGHT      Num       8    F6.2                  WEIGHT
title "Descriptive Statistics for HTWT Data Set";

proc means data=htwt;

run;
                 Descriptive Statistics for HTWT Data Set                  

                             The MEANS Procedure

Variable  Label     N          Mean       Std Dev       Minimum       Maximum

-----------------------------------------------------------------------------

AGE       AGE     237    16.4430380     1.8425767    13.9000000    25.0000000

HEIGHT    HEIGHT  237    61.3645570     3.9454019    50.5000000    72.0000000

WEIGHT    WEIGHT  237   101.3080169    19.4406980    50.5000000   171.5000000

----------------------------------------------------------------------------
title "Oneway Frequency Tabulation for Sex for HTWT Data Set";

proc freq data=htwt;

tables sex;

run;

                         Oneway Frequency Tabulation for Sex for HTWT Data Set                    

                                      The FREQ Procedure

                                                      Cumulative    Cumulative

                 SEX         Frequency     Percent     Frequency      Percent

                 -------------------------------------------------------------

                 f                111       46.84           111        46.84

                 m                126       53.16           237       100.00
ANCOVA Models Using Proc Reg

When we use Proc Reg to fit an ANCOVA model involving interactions, and dummy variables, we must first create these variables in a data step.
In the data step below, we create a dummy variable for FEMALE. We also create CENTAGE, by subtracting 16.5  from each value of AGE. We create two interaction variables, FEM_AGE and FEM_CENTAGE. We will be using these two interaction terms in fitting our ANCOVA models.
/*Create a new data set with new variables*/

data htwt2;

  set htwt;

  /*Create dummy variables for female*/

  if sex="f" then female=1;

  if sex="m" then female=0;  
/*Center age at 16.5 years*/

  centage = age - 16.5;

  /*Create interactions*/

  fem_age = female * age;

  fem_centage = female * centage;

run;

Now we generate a scatter plot with HEIGHT as the Y and AGE as the X , with a separate regression line for males and females.

title "Regression Plot of Height by Age";

title2 "For Males and Females";

proc sgplot data=htwt2;

  where age <= 19;

  reg y=height x=age / group=sex;

run; 

Older participants are taller, for both males and females, but it appears that Males have a slightly lower height at the youngest age, and a steeper slope than females. By the age of 19, males are about five inches taller than females, on average.
ANCOVA Model with Original Variables:

Next, we fit an ANCOVA model, representing the relationships shown on the graph. We include the main effects of sex (FEMALE), AGE, and their interaction. Note that we include the interaction created from the original AGE variable in this model. We again use a where statement to restrict the analysis to those who are less than or equal to 19 years old. We use the clb option to get a 95% confidence interval for each of the parameters in the model.
The model that we are fitting is:

title "ANCOVA for Males and Females";

title2 "Relationship of Height to Age";

proc reg data=htwt2;

  where age <=19;

  model height = female age fem_age / clb;

run; quit;
We interpret the overall significance by looking at the Analysis of Variance table. We see that the model has 3 degrees of freedom, corresponding to the 3 predictors included in the model. The overall model is significant, F(3,215) = 60.93, p<0.001. The adjusted R-square is 0.4520.

                                      Model: MODEL1

                                  Dependent Variable: HEIGHT

                            Number of Observations Read         219

                            Number of Observations Used         219

                                     Analysis of Variance

                                            Sum of           Mean

        Source                   DF        Squares         Square    F Value    Pr > F

        Model                     3     1432.63813      477.54604      60.93    <.0001

        Error                   215     1684.95730        7.83701

        Corrected Total         218     3117.59543

                     Root MSE              2.79947    R-Square     0.4595

                     Dependent Mean       61.00457    Adj R-Sq     0.4520

                     Coeff Var             4.58895

We examine the parameter estimates in the output below. We first look at the parameter estimate for the interaction term. The interaction term, (3, represents the difference in the slope of the regression line for females vs. the reference category (males). The estimated slope of AGE for females is -.92943 less for males. and this difference in slopes is significant (t215 = -3.75, p = .0002). The female rate of growth is about an inch less per year in this age range than the male rate of growth. 
We next look at the parameter estimate for the main effect of AGE. The parameter for AGE , (2, represents the slope for the reference group, males. The estimated slope for males is 2.03, and is significant (t215 = 11.44, p < 0.0001). This means that the average height of males increases by about 2 inches for each year of age. 

The parameter for FEMALE, (1, represents the difference in the intercept for females vs. males, i.e., the difference in predicted average height between females and males when AGE = 0. This is clearly a non-interesting value, because AGE = 0 is outside the range of our data. If we interpret this coefficient, it means that if we extrapolate our regression lines to zero, females are estimated to be 13.6 inches taller than males.
The INTERCEPT, (0, represents the intercept for males (the reference group). The parameter estimate is 28.88, which means we estimate that males would have a height of about 29 inches at age 0, if we extrapolate our regression line to zero. Again, we have no interest in interpreting this intercept.
This whole exercise is a cautionary tale in parameter interpretation. When you have an interaction in a model, it is tricky to interpret the main effects in the model. The main effects have to be interpreted with a great deal of care, if at all, in a model with an interaction. 

                                                     Parameter Estimates

                                Parameter       Standard

           Variable     DF       Estimate          Error    t Value    Pr > |t|       95% Confidence Limits

           Intercept     1       28.88281        2.87343      10.05      <.0001       23.21911       34.54650

           female        1       13.61231        4.01916       3.39      0.0008        5.69031       21.53432

           AGE           1        2.03130        0.17764      11.44      <.0001        1.68117        2.38144

           fem_age       1       -0.92943        0.24782      -3.75      0.0002       -1.41791       -0.44096
ANCOVA Model with Centered Age:
We now refit the ANCOVA model using centered age, CENTAGE, and the interaction between female and centered age, FEM_CENTAGE, as predictors, rather than our original variables.
title "ANCOVA for Males and Females";

title2 "Relationship of Height to Centered Age";

proc reg data=htwt2;

  where age <=19;

  model height = female centage fem_centage / clb;

  plot rstudent. * predicted.;

  output out=outreg p=predict r=resid rstudent=rstud;

run;
Note that the Analysis of Variance table and the model R-Square in the output below are the same as for the previous model. However, the parameter estimates are different.
                                 ANCOVA for Males and Females                                 

                            Relationship of Height to Centered Age

                                       The REG Procedure

                                         Model: MODEL1

                                  Dependent Variable: HEIGHT

                            Number of Observations Read         219

                            Number of Observations Used         219

                                     Analysis of Variance

                                            Sum of           Mean

        Source                   DF        Squares         Square    F Value    Pr > F

        Model                     3     1432.63813      477.54604      60.93    <.0001

        Error                   215     1684.95730        7.83701

        Corrected Total         218     3117.59543

                     Root MSE              2.79947    R-Square     0.4595

                     Dependent Mean       61.00457    Adj R-Sq     0.4520

                     Coeff Var             4.58895

                                            Parameter Estimates

                           Parameter       Standard

    Variable       DF       Estimate          Error    t Value    Pr > |t|       95% Confidence Limits

    Intercept       1       62.39929        0.26902     231.95      <.0001       61.86904       62.92955

    female          1       -1.72336        0.38917      -4.43      <.0001       -2.49044       -0.95629

    centage         1        2.03130        0.17764      11.44      <.0001        1.68117        2.38144

    fem_centage     1       -0.92943        0.24782      -3.75      0.0002       -1.41791

We now take a look at the parameter estimates in detail. We can see that the estimated interaction term, which represents the difference in slope for females vs. males, is the same as in the previous model. Again, we estimate that the slope for females is .93 inches less than the slope for males. The coefficient for CENTAGE, which represents the slope for males (the reference category), is the same as the coefficient for AGE in the previous model, and is 2.03 inches per year. 

However, the estimated values for FEMALE and the INTERCEPT are different than in the previous model. We see that the estimated effect of FEMALE is now -1.72. We can interpret this as the difference in the average height of females vs. males when they are 16.5 years old (i.e. when CENTAGE is zero). That is, females are about 1.7 inches shorter than males, on average, at age 16.5 years, and this difference in height is significant (t215 = -4.43, p < 0.001). This is a meaningful difference, which has a nice interpretation.

The INTERCEPT in this model tells us the estimated average height of males when CENTAGE is zero (meaning that their actual age is 16.5 years). The estimated mean height for males at age 16.5 is 63.4 inches.    

Moral of the story: It is often helpful to center continuous variables in a regression model. It helps in interpreting the intercept in the model, and can also help in interpreting the main effects of variables that are included in interactions.

ANCOVA Model Using Proc GLM

We now refit the model with CENTAGE, but using Proc GLM. The advantage of using this procedure is that we don’t need to create dummy variables for our categorical variables, and the interaction terms do not need to be created in advance. The solution option is used to request that SAS print out the parameter estimates from the model. This option is not necessary, but is used for comparison with the parameter estimates from Proc Reg.

title "ANCOVA model using GLM";

title "Relationship of Height to Centered AGE";

proc glm data=htwt2;

  where age <=19;

  class sex;

  model height = sex centage sex*centage / solution;
run; quit;
                            Relationship of Height to Centered AGE                           

                                       The GLM Procedure

                                   Class Level Information

                                Class         Levels    Values
                                SEX                2    f m

                            Number of Observations Read         219

                            Number of Observations Used         219

Dependent Variable: HEIGHT

                                              Sum of

      Source                      DF         Squares     Mean Square    F Value    Pr > F

      Model                        3     1432.638133      477.546044      60.93    <.0001

      Error                      215     1684.957300        7.837011

      Corrected Total            218     3117.595434

                      R-Square     Coeff Var      Root MSE    HEIGHT Mean

                      0.459533      4.588945      2.799466       61.00457

In the output below, the Type I SS shows the effect of each predictor in the model, sequentially. That is, the effect of SEX is evaluated without controlling for the other predictors. The effect of  AGE is evaluated with only SEX in the model, and the effect of  the CENTAGE by SEX interaction is evaluated, after adjusting for both main effects. The total of the Type I SS is equal to the total model SS.
      Source                      DF       Type I SS     Mean Square    F Value    Pr > F

      SEX                          1       89.225774       89.225774      11.39    0.0009

      centage                      1     1233.182326     1233.182326     157.35    <.0001

      centage*SEX                  1      110.230033      110.230033      14.07    0.0002

The Type III SS below shows the effect of each predictor in the model, controlling for all other effects. The Type III SS is sometimes called the regression sum of squares or partial sum of squares. In this case, the total of the Type III SS does not equal the total model SS.
      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      SEX                          1      153.684358      153.684358      19.61    <.0001

      centage                      1     1252.650148     1252.650148     159.84    <.0001

      centage*SEX                  1      110.230033      110.230033      14.07    0.0002

Notice that we get the same parameter estimates using Proc GLM as we did in Proc Reg. By default, Proc GLM overparameterizes the model, including a parameter for each level of SEX. The parameter estimate for the highest level of SEX is set to zero, which has the effect in this case of making males the reference category, as we had when we fit the model using Proc Reg. Switching the reference categories would not change the overall model fit, but would change the parameter estimates. Although the parameters are not uniquely estimable in this overparameterized model, we can interpret the parameter estimates, knowing the convention that SAS uses for the parameters in the model.

                                                     Standard
            Parameter             Estimate             Error    t Value    Pr > |t|

            Intercept          62.39929305 B      0.26902176     231.95      <.0001

            SEX         f      -1.72336221 B      0.38916785      -4.43      <.0001

            SEX         m       0.00000000 B       .                .         .

            centage             2.03130229 B      0.17763755      11.44      <.0001

            centage*SEX f      -0.92943492 B      0.24782450      -3.75      0.0002

            centage*SEX m       0.00000000 B       .                .         .

NOTE: The X'X matrix has been found to be singular, and a generalized inverse was used to

      solve the normal equations.  Terms whose estimates are followed by the letter 'B' are

      not uniquely estimable.
We now check the plot of studentized residuals vs. predicted values for equality of variances, which we requested as part of the Proc Reg commands:

We can use Proc Univariate to check the distribution of the residuals for normality, and we see that the residuals appear to be quite normal.

Separate Regression Models for Males and Females

We now fit separate regression models for females and males, using CENTAGE as the predictor of HEIGHT for each sex. To do this, we first sort by SEX, and then fit the regression models by SEX, using a by statement. 
And, remembering to select only cases with AGE<=19! We can see from the output below that the parameter estimates for females and males match what we had derived from the ANCOVA models, with Centered Age as a predictor. 
proc sort data = htwt2;

  by sex;
run;
title "Separate Regressions for Females and Males";

proc reg data = htwt2;
  where age <=19;
  by sex;

  model height = centage;

run; quit;
                Separate Regressions for Females and Males                 

----------------------------------- SEX=f -----------------------------------

                              The REG Procedure

                                Model: MODEL1

                      Dependent Variable: HEIGHT HEIGHT

                   Number of Observations Read         103

                   Number of Observations Used         103

                            Analysis of Variance

                                    Sum of          Mean

  Source                  DF       Squares        Square   F Value   Pr > F

  Model                    1     318.63384     318.63384     38.46   <.0001

  Error                  101     836.85005       8.28564

  Corrected Total        102    1155.48388

            Root MSE              2.87848    R-Square     0.2758

            Dependent Mean       60.32718    Adj R-Sq     0.2686

            Coeff Var             4.77145

                             Parameter Estimates

                                Parameter      Standard

 Variable    Label       DF      Estimate         Error   t Value   Pr > |t|

 Intercept   Intercept    1      60.67593       0.28915    209.84     <.0001

 centage                  1       1.10187       0.17768      6.20     <.0001

                 Separate Regressions for Females and Males                

----------------------------------- SEX=m -----------------------------------

                              The REG Procedure

                                Model: MODEL1

                      Dependent Variable: HEIGHT HEIGHT

                   Number of Observations Read         116

                   Number of Observations Used         116

                            Analysis of Variance

                                    Sum of          Mean

  Source                  DF       Squares        Square   F Value   Pr > F

  Model                    1    1024.77852    1024.77852    137.75   <.0001

  Error                  114     848.10725       7.43954

  Corrected Total        115    1872.88578

            Root MSE              2.72755    R-Square     0.5472

            Dependent Mean       61.60603    Adj R-Sq     0.5432

            Coeff Var             4.42741

                             Parameter Estimates

                                Parameter      Standard

 Variable    Label       DF      Estimate         Error   t Value   Pr > |t|

 Intercept   Intercept    1      62.39929       0.26211    238.06     <.0001

 centage                  1       2.03130       0.17307     11.74     <.0001

